preparation (evaluate this cell to initialize)

general definition

definition for F_4

norcharC[1,λ] : computation of $C_\lambda^{(1)}$

norcharC[2,λ] : computation of $C_\lambda^{(2)}$ for $\lambda \in \{0, \omega_1, \omega_2, 2\,\omega_4\}$

norcharD[2,λ] : computation of $D_\lambda^{(2)}$ for $\lambda \in \{0, \omega_1, \omega_2, 2\,\omega_4\}$

proof of Proposition 4.1 (vanishing of sums)

$\lambda = \lambda_2$

```
Block[{ty = F, rk = 4, summand, subgroup, sum},
  summand[w_] := (-1)^Length[w] exp[WeylR[ty, rk][w][rho]] * pfE[w, λ[2]];
  (* subgroup=Select[weylgroup[ty,rk],FreeQ[#,3]&];
  Print[subgroup]; *)
  subgroup = {{}, {1}, {2}, {2, 1}, {1, 2}, {1, 2, 1}} ;
  Print[{"size of the isotropy subgroup : ", subgroup // Length}];
  sum = Total@Map[summand, subgroup];
  sum
] // Simplify
```

```
{size of the isotropy subgroup : , 6}
0
```

$\lambda = \lambda_3$

```
Block[{ty = F, rk = 4, summand, subgroup, sum},
  summand[w_] := (-1)^Length[w] exp[WeylR[ty, rk][w][rho]] * pfE[w, λ[3]];
  (* subgroup=Select[weylgroup[ty,rk],FreeQ[#,2]&]; *)
  subgroup = {{}, {1}, {3}, {3, 1}};
  Print[{"size of the isotropy subgroup : ", subgroup // Length}];
  sum = Total@Map[summand, subgroup];
  sum
] // Simplify
```

```
{size of the isotropy subgroup : , 4}
0
```

$\lambda = \lambda_4$

```
Block[{ty = F, rk = 4, summand, subgroup, sum},
  summand[w_] := (-1)^Length[w] exp[WeylR[ty, rk][w][rho]] * pfE[w, λ[4]];
  subgroup = Select[weylgroup[ty, rk], FreeQ[#, 1] &];
  Print[{"size of the isotropy subgroup : ", subgroup // Length}];
  sum = Total@Map[summand, subgroup];
  sum
] // Simplify
```

```
{size of the isotropy subgroup : , 48}
0
```

$\lambda = \lambda_4$: more efficient check

```
Block[{ty = F, rk = 4, summand, WC3, WC2, sumoverWC2, sum, WC3C2cosets},
 summand[w_] := (-1)^Length[w] exp[WeylR[ty, rk][w][rho]] * pfE[w, λ[4]];
 WC3 = Select[weylgroup[ty, rk], FreeQ[#, 1] &];
 WC2 = Select[WC3, FreeQ[#, 3] &];
 sumoverWC2 = Total@Map[summand, WC2] // Simplify;
 WC3C2cosets = {{}, {3, 2, 3, 4, 3, 2, 3}, {3}, {2, 3, 4, 3, 2, 3}, {4, 3}, {2, 3, 2, 4, 3},
   {2, 3}, {3, 4, 3, 2, 3}, {3, 2, 3}, {4, 3, 2, 3}, {2, 4, 3}, {3, 2, 4, 3}};
 sum = 0;
 Do[
  sum = Simplify[sum + (-1)^Length[ww] (sumoverWC2 /. weyltorule[ty, rk][ww])];
  (* Print[sum]; *)
  , {ww, WC3C2cosets}
 ];
 sum
]
```

```
0
```

proof of $C_\lambda^{(2)} = D_\lambda^{(2)}$ for $\lambda \in \{\omega_1, \omega_2, 2\,\omega_4, 0\}$

$\lambda = \omega_1$

```
Block[{C2, D2, Subscript, la},
  la = wt[1, 0, 0, 0];
  norcharC2C1[la] // Print;
  C2 = norcharC[2, la];
  D2 = norcharD[2, la];
  C2 - D2
 ] // Simplify
```

$$\begin{aligned}
&c[1, wt[0, 1, -2, 2]]\, c[1, wt[1, -1, 2, -2]] \left(1 - \frac{1}{x[2]\, x[4]^2}\right) + \\
&c[1, wt[0, 1, 0, -2]]\, c[1, wt[1, -1, 0, 2]] \left(1 - \frac{x[2]}{x[4]^2}\right) + \\
&c[1, wt[-1, 1, 0, 0]]\, c[1, wt[2, -1, 0, 0]] \left(1 - \frac{1}{x[2]^3\, x[3]^4\, x[4]^2}\right) + \\
&c[1, wt[0, 0, 0, 0]]\, c[1, wt[1, 0, 0, 0]] \left(1 - \frac{1}{x[1]^2\, x[2]^3\, x[3]^4\, x[4]^2}\right) + \\
&c[1, wt[0, -1, 2, 0]]\, c[1, wt[1, 1, -2, 0]] \left(1 - \frac{1}{x[2]\, x[3]^4\, x[4]^2}\right) + \\
&c[1, wt[0, 1, 0, -2]]\, c[1, wt[1, -1, 0, 2]] \left(1 - \frac{x[4]^2}{x[2]}\right) + \\
&c[1, wt[0, 1, -2, 2]]\, c[1, wt[1, -1, 2, -2]] \left(1 - x[2]\, x[4]^2\right) + \\
&c[1, wt[0, -1, 2, 0]]\, c[1, wt[1, 1, -2, 0]] \left(1 - x[2]\, x[3]^4\, x[4]^2\right) + \\
&c[1, wt[-1, 1, 0, 0]]\, c[1, wt[2, -1, 0, 0]] \left(1 - x[2]^3\, x[3]^4\, x[4]^2\right) + \\
&c[1, wt[0, 0, 0, 0]]\, c[1, wt[1, 0, 0, 0]] \left(1 - x[1]^2\, x[2]^3\, x[3]^4\, x[4]^2\right)
\end{aligned}$$

0

$\lambda = \omega_2$

```
Block[{C2, D2, Subscript, la},
  la = wt[0, 1, 0, 0];
  norcharC2C1[la] // Print;
  C2 = norcharC[2, la];
  D2 = norcharD[2, la];
  C2 - D2
 ] // Simplify
```

$$\begin{aligned}
&c[1, wt[-1, 1, 0, 0]]\, c[1, wt[1, 0, 0, 0]] \left(1 - \frac{1}{x[1]}\right) + \\
&c[1, wt[-1, 1, 0, 0]]\, c[1, wt[1, 0, 0, 0]] \left(1 - x[1]\right)
\end{aligned}$$

0

$\lambda = 2\,\omega_4$

```
Block[{C2, D2, Subscript, la},
  la = wt[0, 0, 0, 2];
  norcharC2C1[la] // Print;
  C2 = norcharC[2, la];
  D2 = norcharD[2, la];
  C2 - D2
 ] // Simplify
```

$$c[1, wt[0, -1, 2, 0]]\, c[1, wt[0, 1, -2, 2]] \left(1 - \frac{1}{x[3]^2}\right) +$$
$$c[1, wt[-1, 1, 0, 0]]\, c[1, wt[1, -1, 0, 2]] \left(1 - \frac{1}{x[2]^2\, x[3]^2}\right) +$$
$$c[1, wt[-1, 0, 0, 2]]\, c[1, wt[1, 0, 0, 0]] \left(1 - \frac{1}{x[1]^2\, x[2]^2\, x[3]^2}\right) +$$
$$c[1, wt[0, -1, 2, 0]]\, c[1, wt[0, 1, -2, 2]] \left(1 - x[3]^2\right) +$$
$$c[1, wt[-1, 1, 0, 0]]\, c[1, wt[1, -1, 0, 2]] \left(1 - x[2]^2\, x[3]^2\right) +$$
$$c[1, wt[-1, 0, 0, 2]]\, c[1, wt[1, 0, 0, 0]] \left(1 - x[1]^2\, x[2]^2\, x[3]^2\right)$$

0

$\lambda = 0$

definition :
norcharCList = list of summands in Weyldenom*norcharC
norcharDList = list of summands in Weyldenom*norcharD (equation 4.10)

```
norcharCList[2, wt[0, 0, 0, 0]] = Block[{ty = F, rk = 4, c},
   Map[Factor, Weyldenom * (List @@ (norcharC2C1[wt[0, 0, 0, 0]]) /. {c -> norcharC})]
  ];
norcharDList[2, wt[0, 0, 0, 0]] =
  Block[{ty = F, rk = 4, summand, subgroup134, cosets, sumoversubgroup134},
   summand[w_] := (-1)^Length[w] * exp[WeylR[ty, rk][w][rho]] * pfD[w, wt[0, 0, 0, 0]];
   subgroup134 = Select[weylgroup[ty, rk], FreeQ[#, 2] &];
   sumoversubgroup134 = (Total@Map[summand, subgroup134]) ;
   cosets =
    Map[movetochamberList[ty, rk][#][[1]] &, WeylOrbit[ty, rk][wt[0, 1, 0, 0]]];
   Factor /@ Map[(-1)^Length[#] * sumoversubgroup134 /. weyltorule[ty, rk][#] &, cosets]
];
(* numerical check by specialization*)
(*
-241288326033606549261685390222926122655206829852157088925/235362790369165882420894514010166548594626760011899597463278;
{
 Weyldenom*norcharC[2,wt[0,0,0,0]]/. {x[1]→2,x[2]→3,x[3]→5,x[4]→7},
 Weyldenom*norcharD[2,wt[0,0,0,0]]/. {x[1]→2,x[2]→3,x[3]→5,x[4]→7},
 Total[norcharCList[2,wt[0,0,0,0]]/. {x[1]→2,x[2]→3,x[3]→5,x[4]→7}],
 Total[norcharDList[2,wt[0,0,0,0]]/.{x[1]→2,x[2]→3,x[3]→5,x[4]→7}]
}
*)
```

proof

```
(* this function is to group a list of rational functions
 in x_i by its denominator with x_i specialized by numbers *)
maxprimefactor[xx_, num_List: {2, 3, 5, 7}] := Block[{temp, rule},
  rule = Inner[Rule, Array[x, 4], num, List];
  temp = xx /. rule;
  temp = Denominator@temp;
  temp = FactorInteger@temp;
  temp = (First /@ temp);
  Max@temp
 ]
Block[{differencetemp, dlist, timestart, numtogo, num, repeatGather},
 Label["begin"]; Print["process initiated!"];
 {num, repeatGather} = {{3, 7, 2, 5}, 3};
 dlist = norcharDList[2, wt[0, 0, 0, 0]];
 Do[
  dlist = Map[Total, GatherBy[dlist, maxprimefactor[#, num] &]];
  , {repeatGather}
 ];
 dlist = SortBy[dlist, maxprimefactor[#, num] &] // Reverse;
 dlist = Map[Total[dlist[[#]]] &, {{1}, {2, 3, 4}, {5}, {6},
    {7, 8, 9}, {10}, {11}, {12, 13, 14, 15}, {16, 17, 18, 19, 20, 21, 22}}];
 timestart = AbsoluteTime[];
 numtogo = Length[dlist];
 differencetemp = Factor[Total[norcharCList[2, wt[0, 0, 0, 0]]]];
 Print[{"step number", "time elapsed", "size of the expression"}];
 Do[
  differencetemp = Factor[differencetemp - dlist[[s]]];
  Print[{ToString[s] <> "/" <> ToString[numtogo],
    Floor[AbsoluteTime[] - timestart], LeafCount[differencetemp]}],
  {s, 1, numtogo}
 ];
 differencetemp
]
```

```
process initiated!

{step number, time elapsed, size of the expression}

{1/9, 70, 5610568}

{2/9, 261, 11298124}

{3/9, 433, 10849511}

{4/9, 493, 11839177}

{5/9, 889, 13922778}

{6/9, 1220, 12291439}

{7/9, 1295, 11109440}

{8/9, 1445, 3039509}

{9/9, 1455, 1}

0
```